
pkgname=doublecmd-plugin-adb
_pkgname=adbplugin-x64
pkgver=8.8
pkgrel=1
pkgdesc="The ultimate ADB (Android Debug Bridge) file system plugin for Double Commander and Total Commander with extra features"
arch=(x86_64)
url="http://www.uniqtec.eu/applications/android-adb.html"
license=('unknown')
depends=('doublecmd' 'android-tools' 'pcre')
makedepends=('patchelf')
options=(!strip)
source=('https://www.uniqtec.eu/download/adb/adbplugin-linux-x64.zip')
md5sums=('71be0815cb0df952c4e617b745f64860')

package() {
  patchelf --replace-needed libpcre.so.3 libpcre.so adb_rename.wfx
  install -d "${pkgdir}/usr/lib/doublecmd/plugins/wfx/${_pkgname}"
  install -d "${pkgdir}/usr/lib/doublecmd/plugins/wfx/${_pkgname}/doc"

  install -m 644 adb_rename.wfx "${pkgdir}/usr/lib/doublecmd/plugins/wfx/${_pkgname}/adb.wfx64"
  install -m 644 doc/notes.txt "${pkgdir}/usr/lib/doublecmd/plugins/wfx/${_pkgname}/doc"
  install -m 644 adb.ico "${pkgdir}/usr/lib/doublecmd/plugins/wfx/${_pkgname}"
}
